

Pony Truss ASD



F_y := 360·MPa

E := 200000·MPa

ν := 0.3

SF := 2.12

from 1.50 it would suffice it is indicated in the text
I don't think so would desprend from code

Truss Data

N_{panels} := 12

L₁ := 4·m

L_{ww} := N_{panels}·L₁

L = 48 m

Floor Beam Data

L_B := 9·m

h := 68·cm

t_w := 12·mm

b_f := 25·cm

t_f := 16·mm

Verticals Data

L_C := 3·m

H_{ww} := 25·cm

T_w := 8·mm

B_f := 20·cm

T_f := 11·mm

P := 170·ton

service level maximum compression on the compression chord

seismic := 1

1 for strong earthquake
0 for weak or no

Moment in the transverse plane for the check of the vertical

$M_C := 0.3 \cdot \frac{\text{kip}}{\text{ft}} \cdot L_1 \cdot L_C$

M_C = 5.36 m·ton

check the vertical for this service level moment plus tension or
compression (depending upon web setup) from general structure

M_{C_u} := SF·M_C

M_{C_u} = 11.36 m·ton

same but factored at SF

K factor (on panel length L₁) for the compression chord following the Holt procedure



$$I_C := B_f \cdot \frac{H^3}{12} - \left(B_f - T_w \right) \cdot \frac{\left(H - 2 \cdot T_f \right)^3}{12} \qquad I_B := b_f \cdot \frac{h^3}{12} - \left(b_f - t_w \right) \cdot \frac{\left(h - 2 \cdot t_f \right)^3}{12}$$

It is assumed both verticals and floor beam bend in the transversal plane in strong axis

$$C_{\text{w}} := \frac{E}{I_C^2 \cdot \left(\frac{L_C}{3 \cdot I_C} + \frac{L_B}{2 \cdot I_B} \right)} \qquad C = 1.26 \frac{\text{ton}}{\text{cm}}$$

$$P_c := SF \cdot P$$

$$\beta := \frac{\frac{C}{\frac{\text{kip}}{\text{in}}} \cdot \frac{L_1}{\text{in}}}{\frac{P_c}{\text{kip}}}$$

Now we build the interpolation first by columns, then by horizontals

X4 :=

1

.9

.8

.7

.6

.5

.4

.3

.293

Y4 :=

3.686

3.352

2.961

2.448

2.035

1.75

1.232

0.121

0

X6 :=

1

.98

.96

.94

.92

.9

.85

.8

.75

.7

.65

.6

.55

.5

.45

.4

.35

.3

.259

Y6 :=

3.616

3.284

3

2.754

2.643

2.593

2.46

2.313

2.147

1.955

1.739

1.639

1.517

1.362

1.158

.886

.53

.187

0

X8 :=

1

.98

.96

.95

.9

.85

.8

.75

.7

.65

.6

.55

.5

.45

.4

.35

.3

.25

.2

.18

Y8 :=

3.66

2.944

2.665

2.595

2.263

2.013

1.889

1.75

1.595

1.442

1.338

1.211

1.047

0.829

0.627

0.434

0.249

0.135

0.045

0

X10 :=	1	Y10 :=	3.714	X12 :=	1	Y12 :=	3.754	X14 :=	1	Y14 :=	3.785
	0.98		2.806		0.98		2.787		0.98		2.771
	0.96		2.542		0.96		2.456		0.96		2.454
	0.94		2.303		0.94		2.252		0.94		2.254
	0.92		2.146		0.92		2.094		0.92		2.101
	0.9		2.045		0.9		1.951		0.9		1.968
	0.85		1.794		0.85		1.709		0.85		1.681
	0.8		1.629		0.8		1.48		0.8		1.456
	0.75		1.501		0.75		1.344		0.75		1.273
	0.7		1.359		0.7		1.2		0.7		1.111
	0.65		1.236		0.65		1.087		0.65		0.988
	0.6		1.133		0.6		0.985		0.6		0.878
	0.55		1.007		0.55		0.86		0.55		0.768
	0.5		0.847		0.5		0.75		0.5		0.668
	0.45		0.714		0.45		0.624		0.45		0.537
	0.4		0.555		0.4		0.454		0.4		0.428
	0.35		0.352		0.35		0.323		0.35		0.292
	0.3		0.17		0.3		0.203		0.3		0.183
	0.25		0.107		0.25		0.103		0.25		0.121
	0.2		0.068		0.2		0.055		0.2		0.053
	0.15		0.017		0.15		0.031		0.15		0.029
	0.139		0		0.114		0		0.097		0

	1		3.809		1		4
	0.98		2.774		0.98		3.73
	0.96		2.479		0.96		3.478
	0.94		2.282		0.94		3.244
	0.92		2.121		0.92		3.026
	0.9		1.981		0.9		2.822
	0.85		1.694		0.85		2.372
	0.8		1.465		0.8		1.993
	0.75		1.262		0.75		1.673
	0.7		1.088		0.7		1.401
	0.65		0.94		0.65		1.169
X16 :=	0.6	Y16 :=	0.808	X32 :=	0.6	Y32 :=	0.97
	0.55		0.708		0.55		0.798
	0.5		0.6		0.5		0.648
	0.45		0.5		0.45		0.519
	0.4		0.383		0.4		0.406
	0.35		0.28		0.35		0.309
	0.3		0.187		0.3		0.226
	0.25		0.112		0.25		0.157
	0.2		0.07		0.2		0.1
	0.15		0.025		0.15		0.056
	0.1		0.01		0.1		0.025
	0.085		0		0		0

```

X4 := reverse(X4)    X6 := reverse(X6)    X8 := reverse(X8)    X10 := reverse(X10)    X12 := reverse(X12)

Y4 := reverse(Y4)    Y6 := reverse(Y6)    Y8 := reverse(Y8)    Y10 := reverse(Y10)    Y12 := reverse(Y12)

X14 := reverse(X14)  X16 := reverse(X16)  X32 := reverse(X32)

Y14 := reverse(Y14)  Y16 := reverse(Y16)  Y32 := reverse(Y32)

```

```
vs4 := cspline(X4, Y4)      v4(y) := interp(vs4, X4, Y4, y)

vs6 := cspline(X6, Y6)      v6(y) := interp(vs6, X6, Y6, y)

vs8 := cspline(X8, Y8)      v8(y) := interp(vs8, X8, Y8, y)

vs10 := cspline(X10, Y10)   v10(y) := interp(vs10, X10, Y10, y)

vs12 := cspline(X12, Y12)   v12(y) := interp(vs12, X12, Y12, y)

vs14 := cspline(X14, Y14)   v14(y) := interp(vs14, X14, Y14, y)

vs16 := cspline(X16, Y16)   v16(y) := interp(vs16, X16, Y16, y)

vs32 := cspline(X32, Y32)   v32(y) := interp(vs32, X32, Y32, y)
```

now we interpolate along the horizontal lines

$$XX := \begin{pmatrix} 4 \\ 6 \\ 8 \\ 10 \\ 12 \\ 14 \\ 16 \\ 32 \end{pmatrix}$$

substitute at left in the last position the (even) number of
panels you admit the Lutz-Fisher column represents

$$YY(y) := \begin{pmatrix} v4(y) \\ v6(y) \\ v8(y) \\ v10(y) \\ v12(y) \\ v14(y) \\ v16(y) \\ v32(y) \end{pmatrix}$$

```
vvs(y) := cspline(XX, YY(y))
```

```
KK(x, y) := interp(vvs(y), XX, YY(y), x)    this function then resumes one  
                                              interpolation fitting the data
```

You see our problem is an inverse one, and must find what we have called y
corresponds to β when the number of panels is x. Since the automatical solving
feature does not yield proper result, we proceed manually



$K_{\text{var}} := 1.29718$

vary repeatedly the buckling factor till the quotient below yields 1

$$\frac{KK\left(N_{\text{panels}}, \frac{1}{K}\right)}{\beta} = 1$$

$KK\left(N_{\text{panels}}, \frac{1}{K}\right) = 1.3951$

$\beta = 1.3951$

when the values in orange are equal you may accept you have found the correct K

Optimum design of the compression chord member as a square box in ASD

Assume a square box shape to be dimensioned by ASD

$b_{\text{max}} := 100\cdot\text{cm}$

$b_{\text{min}} := 10\cdot\text{cm}$

$t_{\text{max}} := 5\cdot\text{cm}$

$t_{\text{min}} := 3\cdot\text{mm}$



$b := b_{\text{min}} \qquad t := t_{\text{min}} \qquad \text{unwarranted guesses}$

$A(b,t) := b^2 - (b - 2\cdot t)^2 \qquad I(b,t) := \frac{b^4}{12} - \frac{(b - 2\cdot t)^4}{12} \qquad r(b,t) := \sqrt{\frac{I(b,t)}{A(b,t)}}$

$\lambda(b,t) := \frac{K\cdot L_1}{r(b,t)} \qquad \lambda(b,t) = 130.97$

$C_c := \pi\cdot\sqrt{\frac{2\cdot E}{F_y}} \qquad C_c = 104.72$

$$F_a(b,t) := F_y \cdot \left| \begin{array}{l} \frac{1 - \frac{1}{2} \cdot \left(\frac{\lambda(b,t)}{C_c} \right)^2}{\frac{5}{3} + \frac{3}{8} \cdot \frac{\lambda(b,t)}{C_c} - \frac{1}{8} \cdot \left(\frac{\lambda(b,t)}{C_c} \right)^3} \text{ if } \lambda(b,t) \leq C_c \\ \frac{12 \cdot \pi^2 \cdot \frac{E}{F_y}}{23 \cdot (\lambda(b,t))^2} \text{ otherwise} \end{array} \right.$$

$$P_{max1}(b,t) := 0.85 \cdot 0.85 \cdot F_y \cdot A(b,t) \qquad P_{max2}(b,t) := 0.5 \cdot F_a(b,t) \cdot 0.85 \cdot SF \cdot A(b,t)$$

$$P_{max}(b,t) := \left| \begin{array}{l} P_{max2}(b,t) \text{ if seismic} = 1 \\ P_{max1}(b,t) \text{ otherwise} \end{array} \right.$$

$$\lambda_{max} := \left| \begin{array}{l} \frac{110}{\sqrt{\frac{F_y}{ksi}}} \text{ if seismic} = 1 \\ \frac{238}{\sqrt{\frac{F_y}{ksi}}} \text{ otherwise} \end{array} \right.$$

$$Weight(b,t) := A(b,t) \cdot 1 \cdot m \cdot 7850 \cdot \frac{kgf}{m^3} \cdot \frac{1}{m}$$

Given

$F_a(b,t) \cdot A(b,t) \geq P$

$b \geq b_{\min}$

$b \leq b_{\max}$

$t \geq t_{\min}$

$t \leq t_{\max}$

$\frac{b}{t} \leq \lambda_{\max}$

$P_c \leq P_{\max}(b,t)$



$\begin{pmatrix} \textcolor{green}{b} \\ \textcolor{green}{t} \end{pmatrix} := \text{Minimize}(A,b,t)$

$b = 29.86\text{ cm}$

$t = 1.96\text{ cm}$

$A(b,t) = 218.84\text{ cm}^2$

$\text{Weight}(b,t) = 171.79 \frac{\text{kgf}}{\text{m}}$

$\frac{\frac{b}{t}}{\lambda_{\max}} = 1$

$F_a(b,t) = 0.5F_y$

$F_a(b,t) \cdot A(b,t) = 400\text{ ton}$

allowable capacity may exceed much that required if seismic